Vault Guardians Audit Report

Version 0.1

November 19, 2025

Vault Guardians Audit Report

mhng

November 19, 2025

Vault Guardians Audit Report

Prepared by: mhng Lead Auditors: mhng

- mhng

Assisting Auditors:

- None

Table of contents

See table

    - [L-7] A GuardianAdded event is emitted before the vault creation is confirmed leading to a false positive in the off-chain information flow
    - [L-8] Incorrect Name and Symbol when creating tokenVault for i_TokenTwo
    - [L-9] Return value not assigned to named return variable in _aaveDivest
    - [L-10] Missing check for sufficient liquidity before calling removeLiquidity()
    - [L-11] Misleading/Incorrect event emision in updateGuardianAndDaoCut
    - [L-12] Ambiguous guardian and DAO cut calculation can lead to misleading configuration
    - [L-13] Missing event emission on user deposit
- Informational
    - [I-1] Unused interfaces IInvestableUniverseAdapter and IVaultGuardians with commented-out functions
    - [I-2] Insufficient test coverage
    - [I-3] No check on return value in _aaveDivest
    - [I-4] Typo in function title in getUniswapLiquidtyToken
    - [I-5] Missing mechanism for adding new investment platforms
- Gas
    - [G-1] Functions marked public are not used internally

About mhng

Disclaimer

We are making all efforts to find as many vulnerabilities in the code in the given time period, but we are not holding any responsibilities for the the findings provided in this document. A security audit is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
	High	H	H/M	M

		Impact		
Likelihood	Medium	H/M	M	M/L
	Low	M	M/L	L

Audit Details

The findings described in this document correspond the following commit hash:

```
0600272180b6b6103f523b0ef65b070064158303
```

Scope

```
./src/
#-- abstract
|   #-- AStaticTokenData.sol
|   #-- AStaticUSDCData.sol
|   #-- AStaticWethData.sol
#-- dao
|   #-- VaultGuardianGovernor.sol
|   #-- VaultGuardianToken.sol
#-- interfaces
|   #-- IVaultData.sol
|   #-- IVaultGuardians.sol
|   #-- IVaultShares.sol
|   #-- InvestableUniverseAdapter.sol
#-- protocol
|   #-- VaultGuardians.sol
|   #-- VaultGuardiansBase.sol
|   #-- VaultShares.sol
|   #-- investableUniverseAdapters
|       #-- AaveAdapter.sol
|       #-- UniswapAdapter.sol
#-- vendor
    #-- DataTypes.sol
    #-- IPool.sol
    #-- IUniswapV2Factory.sol
    #-- IUniswapV2Router01.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- Tokens:

- weth: https://etherscan.io/token/0xc02aaa39b223fe8d0a0e5c4f27ead9083c756cc2
- link: https://etherscan.io/token/0x514910771af9ca656af840dff83e8264ecf986ca
- usdc: https://etherscan.io/token/0xa0b86991c6218b36c1d19d4a2e9eb0ce3606eb48

Protocol Summary

This protocol allows users to deposit certain ERC20s into an ERC4626 vault managed by a human being, or a `vaultGuardian`. The goal of a `vaultGuardian` is to manage the vault in a way that maximizes the value of the vault for the users who have despoited money into the vault.

You can think of a `vaultGuardian` as a fund manager.

To prevent a vault guardian from running off with the funds, the vault guardians are only allowed to deposit and withdraw the ERC20s into specific protocols.

- Aave v3
- Uniswap v2
- None (just hold)

These 2 protocols (plus "none" makes 3) are known as the "investable universe".

The guardian can move funds in and out of these protocols as much as they like, but they cannot move funds to any other address.

The goal of a vault guardian, is to move funds in and out of these protocols to gain the most yield. Vault guardians charge a performance fee, the better the guardians do, the larger fee they will earn.

Anyone can become a Vault Guardian by depositing a certain amount of the ERC20 into the vault. This is called the `guardian stake`. If a guardian wishes to stop being a guardian, they give out all user deposits and withdraw their guardian stake.

Users can then move their funds between vault managers as they see fit.

The protocol is upgradeable so that if any of the platforms in the investable universe change, or we want to add more, we can do so.

User flow

1. User deposits an ERC20 into a guardian's vault
2. The guardian automatically move the funds based on their strategy
3. The guardian can update the settings of their strategy at any time and move the funds
4. To leave the pool, a user just calls `redeem` or `withdraw`

The DAO

Guardians can earn DAO tokens by becoming guardians. The DAO is responsible for: - Updating pricing parameters - Getting a cut of all performance of all guardians

Summary

Users can stake some ERC20s to become a vault guardian. Other users can allocate them funds in order to maximize yield. The guardians can move the funds between Uniswap, Aave, or just hold the funds. The guardians are incentivized to maximize yield, as they earn a performance fee.

Roles

- User: Someone that wants to deposit assets in a vault to maximize yield
- Guardian: Someone that stakes an ERC20 amount to become a Vault Guardian by creating a new vault and are incentivized to maximize yield, by earning a fee.
- Vault Guardian DAO: Organization that controls Vault operations and earns a fee as incentive

Executive Summary

Issues found

Severity	Number of issues found
High	9
Medium	6
Low	13
Info	5
Gas	1
Total	34

Findings

High

[H-1] totalAssets() ignores all invested funds which leads to severe disruptions of protocol functionality

Description: OpenZeppelin's ERC4626 implementation of `totalAssets()` only checks for the vault balance, not taking into consideration invested amounts, which left the vault.

```
function totalAssets() public view virtual returns (uint256) {
    return _asset.balanceOf(address(this));
}
```

Vault moves 50% of user funds out into: Aave (25%) + Uniswap LP (25%)

`totalAssets()` will return just 50% of the real correct value.

Impact:

Other functions rely on `totalAssets()` return value to calculate different return values:

- `convertToShares`
- `convertToAssets`
- `previewWithdraw`
- `previewRedeem`
- `maxWithdraw`
- `maxRedeem`
- `withdraw`
- `deposit`

Returning an incorrect value will have a severe economic and functional impact on the protocol:

1. Users receive too many shares on deposit, if `totalAssets()` is too small (because invested assets are ignored), then shares become extremely cheap → users get overminted.
2. Withdrawals and redeem become corrupted because they use `convertToShares` and `convertToAssets`, which use `totalAssets()`
3. Governance votes using share balances become inaccurate

Probably more…

Proof of Concept:

- Example scenario:

1. User deposits 100 ether
2. 50 ether gets automatically invested in Aave(25 ether) + Uniswap(25 ether)
3. `totalAssets()` is called as consequence of a protocol action
4. `totalAssets()` returns only 50% of the total amount since the vault only accounts for the holding allocation(50%)
5. Calculations are erroneous, state is incorrect and functionality is disrupted.

Proof of Code:

- Add this test in test/unit/concrete/VaultGuardiansBaseTest.t.sol:

Code

```
    function testTotalAssetsReturnsIncorrectAmount()
            public
            hasGuardian
            hasTokenGuardian
        {
            // user deposits usdc
            usdc.mint(mintAmount, user);
            vm.startPrank(user);
            usdc.approve(address(usdcVaultShares), mintAmount);
            usdcVaultShares.deposit(mintAmount, user);
            vm.stopPrank();

            uint256 expectedVaultBalance = mintAmount;
            uint256 actualVaultBalance = usdcVaultShares.totalAssets();
            uint256 delta = (expectedVaultBalance * 6) / 10; // missing deposit amount approx. 60% -> 25% Aave + 25% Uniswap + ~10%(fees + dilution)

            console.log("Expected Vault Balance: ", expectedVaultBalance);
            console.log("Actual Vault Balance: ", actualVaultBalance);
            console.log("Expected difference in vault amount: ", delta);
            console.log(
                "Actual difference in vault amount: ",
                expectedVaultBalance - actualVaultBalance
            );

            // Shows that there is a significant difference between expected and actual vault amount, approx. 60%
            assertApproxEqAbs(actualVaultBalance, expectedVaultBalance, delta);
        }
```

Recommended Mitigation:

1. Override OpenZepelin's ERC4626 `totalAssets()` implementation to include investment amounts
2. Implement a custom function to keep track of total assets in the vault and don't rely on OpenZeppelin's solution

[H-2] Missing decimal normalization causes reverts when supplying non 18 decimal tokens (USDC) to Aave

Description:

In `_aaveInvest()`, the vault passes `amount` directly into Aave's `supply()` call:

```
    function _aaveInvest(IERC20 asset, uint256 amount) internal {
        bool succ = asset.approve(address(i_aavePool), amount);
        if (!succ) {
            revert AaveAdapter__TransferFailed();
        }
        i_aavePool.supply({
            asset: address(asset),
            amount: amount,
            onBehalfOf: address(this), // decides who gets Aave's
                aTokens for the investment. In this case, mint it to the
                 vault
            referralCode: 0
        });
    }
```

The vault's logic assumes 18-decimal normalized math internally, but Aave expects native token units. This means tokens like USDC (6 decimals) are handled incorrectly:

Vault user deposits: 1e6 USDC (1.0 USDC normalized to 18 decimals = 1e18)

Adapter calls supply(asset, 1e18, ...)

Aave interprets that as 1,000,000,000,000 USDC (1e12 USDC), which far exceeds the user's balance.

Aave's internal ERC20 transferFrom() then reverts due to insufficient balance or allowance.

Impact:

- The vault cannot invest or rebalance when using 6-decimal tokens such as USDC(or USDT).
- Every `_aaveInvest()` call will revert on `i_aavePool.supply()`, blocking deposits, rebalances, or withdrawals that trigger divest/reinvest.
- This blocks all vault operations for such assets resulting in DoS.

Proof of Concept:

- Scenario step by step:

1. Vault uses USDC as its asset and a user tries to deposit 1 USDC (1e6), the vault may call:

_aaveInvest(IERC20(USDC), 1e18);

2. Aave receives:

supply(asset=USDC, amount=1e18)

3. Aave executes internal call:

USDC.transferFrom(vault, aUSDC, 1_000_000_000_000)

4. Call to transferFrom reverts with:

Error: ERC20: transfer amount exceeds balance

Proof of Code for calling redeem():

- Add this code in a new file in test/ProofOfCodes:

Code

```
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.20;

import "forge-std/Test.sol";
import {AaveAdapter} from "../../src/protocol/
    InvestableUniverseAdapters/AaveAdapter.sol";
import {IPool} from "../../src/vendor/IPool.sol";
import {IERC20} from "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import {IERC20Metadata} from "@openzeppelin/contracts/token/ERC20/
    extensions/IERC20Metadata.sol";
import {ERC20} from "@openzeppelin/contracts/token/ERC20/ERC20.sol";

// Mock USDC token with 6 decimals
contract MockUSDC is ERC20 {
    constructor() ERC20("Mock USDC", "USDC") {}

    function decimals() public pure override returns (uint8) {
        return 6;
    }

    function mint(address to, uint256 amount) external {
        _mint(to, amount);
    }
}

// Mock Aave pool
contract MockAavePool {
```

```
26      function supply(
27          address asset,
28          uint256 amount,
29          address onBehalfOf,
30          uint16
31      ) external {
32          IERC20(asset).transferFrom(onBehalfOf, address(this), amount);
33      }
34  }
35
36  // Mock vault with Aave Adapter for investment strategy
37  contract MockVault is AaveAdapter {
38      IERC20 public immutable asset;
39
40      constructor(address pool, IERC20 _asset) AaveAdapter(pool) {
41          asset = _asset;
42      }
43
44      function invest(uint256 amount) external {
45          // Simulate deposit of tokens into the vault
46          _aaveInvest(asset, amount);
47      }
48  }
49
50  // ---------------------------
51  //       Test Contract
52  // ---------------------------
53
54  contract AaveDecimalsMismatchNon18TokensPoC is Test {
55      MockUSDC private usdc;
56      MockAavePool private aavePool;
57      MockVault private vault;
58
59      error ERC20InsufficientBalance(address, uint256, uint256);
60
61      function setUp() public {
62          usdc = new MockUSDC();
63          aavePool = new MockAavePool();
64          vault = new MockVault(address(aavePool), usdc);
65
66          usdc.mint(address(vault), 1e6);
67      }
68
69      function test_USDCRevertsDueToDecimalMismatch() public {
70          // Simulate normalized 18-decimal math, vault believes amount
                is 1e18
71          uint256 normalAmount = 1e18;
72
73          uint256 currentBalance = usdc.balanceOf(address(vault));
74          console.log("Current Balance:", currentBalance);
75
```

```
76          uint256 requiredBalance = normalAmount;
77          console.log("Required Balance:", requiredBalance);
78
79          // Expect revert because trying to transfer 1e18 USDC
80          vm.expectRevert(
81              abi.encodeWithSelector(
82                  ERC20InsufficientBalance.selector,
83                  address(vault),
84                  currentBalance,
85                  requiredBalance
86              )
87          );
88
89          vault.invest(normalAmount);
90      }
91  }
```

Recommended Mitigation:

Add proper decimal scaling before interacting with Aave:

```
1       function _aaveInvest(IERC20 asset, uint256 amount) internal {
2   +       uint8 assetDecimals = IERC20Metadata(address(asset)).decimals()
        ;
3   +       uint256 scaledAmount = amount;
4
5   +       if (assetDecimals < 18) {
6   +           scaledAmount = amount / (10 ** (18 - assetDecimals));
7   +       } else if (assetDecimals > 18) {
8   +           scaledAmount = amount * (10 ** (assetDecimals - 18));
9   +       }
10
11          bool succ = asset.approve(address(i_aavePool), amount);
12          if (!succ) {
13              revert AaveAdapter__TransferFailed();
14          }
15          i_aavePool.supply({
16              asset: address(asset),
17  +           amount: scaledAmount,
18              onBehalfOf: address(this),
19              referralCode: 0
20          });
21      }
```

[H-3] Missing counterparty token approvals in _uniswapDivest result in redeem DoS and locked funds

Description:

The function _uniswapDivest() calls removeLiquidity() and swapExactTokensForTokens() without approving the Uniswap router to transfer the required LP or counterparty tokens. Both Uniswap functions rely on transferFrom under the hood, which reverts if the router has no allowance. As a result, any attempt to divest liquidity or redeem user funds will revert.

Impact:

The vault cannot divest or redeem positions because _uniswapDivest() always reverts ==> DoS + locked funds

Proof of Concept:

Without the approvals for LP and counterparty tokens, calling redeem() or rebalanceFunds(), which both call _uniswapDivest will result in a revert with `ERC20InsufficientAllowance`.

Proof of Code for calling redeem():

- Add this code in a new file in test/ProofOfCodes:

Code

```
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.20;

import "forge-std/Test.sol";
import {VaultShares} from "../../src/protocol/VaultShares.sol";
import {IVaultShares, IVaultData} from "../../src/interfaces/
    IVaultShares.sol";
import {ERC20Mock} from "../mocks/ERC20Mock.sol";
import {DataTypes} from "../../src/vendor/DataTypes.sol";

// ---------------------------
//        Uniswap mocks
// ---------------------------

contract UniswapLP is ERC20Mock {
    // Inherit ERC20Mock: free mint/burn helpers already available
}

contract UniswapFactory {
    address public immutable pair;

    constructor(address pair_) {
        pair = pair_;
    }

    function getPair(address, address) external view returns (address)
        {
        return pair;
    }
```

```
}

// This router is Uniswap-V2-shaped enough for VaultShares. UniswapAdapter
contract UniswapRouter {
    address private _factory;
    UniswapLP private _lp;

    constructor(address factory_, address lp_) {
        _factory = factory_;
        _lp = UniswapLP(lp_);
    }

    function factory() external view returns (address) {
        return _factory;
    }

    function swapExactTokensForTokens(
        uint256 amountIn,
        uint256 /* amountOutMin */,
        address[] calldata path,
        address to,
        uint256 /* deadline */
    ) external returns (uint256[] memory amounts) {
        // pull input
        ERC20Mock(path[0]).transferFrom(msg.sender, address(this), amountIn);

        // use mint to avoid needing pre-funding
        ERC20Mock(path[1]).mint(amountIn, to);

        amounts = new uint256[](path.length);
        amounts[0] = amountIn;
        amounts[1] = amountIn;
    }

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint256 amountADesired,
        uint256 amountBDesired,
        uint256 /* amountAMin */,
        uint256 /* amountBMin */,
        address to,
        uint256 /* deadline */
    ) external returns (uint256 amountA, uint256 amountB, uint256 liquidity) {
        // pull all desired tokens from caller
        ERC20Mock(tokenA).transferFrom(
            msg.sender,
            address(this),
```

```
            amountADesired
        );
        ERC20Mock(tokenB).transferFrom(
            msg.sender,
            address(this),
            amountBDesired
        );

        _lp.mint(100 ether, to);

        return (amountADesired, amountBDesired, 100 ether);
    }

    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint256 liquidity,
        uint256 /* amountAMin */,
        uint256 /* amountBMin */,
        address to,
        uint256 /* deadline */
    ) external returns (uint256 amountA, uint256 amountB) {
        // take the LP from caller
        _lp.transferFrom(msg.sender, address(this), liquidity);
        // burn
        _lp.burn(liquidity, address(this));

        // return *1 wei* of each token to the vault
        ERC20Mock(tokenA).mint(1, to);
        ERC20Mock(tokenB).mint(1, to);

        return (1, 1);
    }
}

// ---------------------------
//         Aave mocks
// ---------------------------

// Deploy a small MockAavePool that returns (aTokenAddress, 0,0,...)
contract MockAavePool {
    address public immutable aTokenAddr;

    constructor(address _aTokenAddr) {
        aTokenAddr = _aTokenAddr;
    }

    // Must match the interface signature used by VaultShares
    function getReserveData(
        address
    ) external view returns (DataTypes.ReserveData memory r) {
```

```
127         // leave everything at the zero default, just set aTokenAddress
128         r.aTokenAddress = aTokenAddr;
129     }
130
131     function supply(
132         address /* asset */,
133         uint256 amount,
134         address onBehalfOf,
135         uint16 /* referralCode */
136     ) external {
137         // For the PoC we don't simulate any lending, just acknowledge
                the call.
138         if (amount > 0) {
139             ERC20Mock(aTokenAddr).mint(amount, onBehalfOf);
140         }
141     }
142 }
143
144 // ---------------------------
145 //       Test Contract
146 // ---------------------------
147
148 contract MissingApprovalsUniswapDivestPoC_Test is Test {
149     VaultShares private vault;
150     ERC20Mock private asset;
151     ERC20Mock private tokenOne;
152     ERC20Mock private mockWETH; // distinct mock WETH
153     ERC20Mock private awethTokenMock;
154
155     UniswapRouter router;
156     UniswapLP lp;
157
158     error ERC20InsufficientAllowance(address, uint256, uint256);
159
160     function setUp() public {
161         asset = new ERC20Mock();
162         tokenOne = new ERC20Mock();
163         mockWETH = new ERC20Mock(); // distinct mock WETH
164         awethTokenMock = new ERC20Mock();
165
166         MockAavePool pool = new MockAavePool(address(awethTokenMock));
167
168         // LP + Factory + Router
169         lp = new UniswapLP();
170         UniswapFactory factory = new UniswapFactory(address(lp));
171         router = new UniswapRouter(address(factory), address(lp));
172
173         // Build constructor data
174         IVaultShares.ConstructorData memory c = IVaultShares.
                ConstructorData({
175             asset: asset,
```

```
176             vaultName: "PoC Vault",
177             vaultSymbol: "POC",
178             guardian: address(this),
179             allocationData: IVaultData.AllocationData({
180                 holdAllocation: 0,
181                 uniswapAllocation: 1000,
182                 aaveAllocation: 0
183             }),
184             aavePool: address(pool),
185             uniswapRouter: address(router),
186             guardianAndDaoCut: 100,
187             vaultGuardians: address(this),
188             weth: address(mockWETH),
189             usdc: address(tokenOne)
190         });
191
192         vault = new VaultShares(c);
193
194         // Label for nicer traces
195         vm.label(address(vault), "VaultShares(POC)");
196         vm.label(address(asset), "ASSET");
197         vm.label(address(tokenOne), "TOKEN_ONE");
198         vm.label(address(factory), "Factory");
199         vm.label(address(router), "Router");
200         vm.label(address(lp), "LP");
201     }
202
203     function test_MissingApprovalsUniswapDivestResultsInRevert() public
            {
204         // User mints and deposits 100 ASSET
205         address user = address(0xBEEF);
206         uint256 depositAmt = 100 ether;
207
208         asset.mint(depositAmt, user);
209
210         // pre-fund the vault so the double-accounting won't cause
               insufficient balance
211         // this must happen after vault creation (setUp) but before
               deposit triggers invest()
212         asset.mint(depositAmt / 2, address(vault));
213
214         // Approve & deposit
215         vm.startPrank(user);
216         asset.approve(address(vault), depositAmt);
217         vault.deposit(depositAmt, user);
218         vm.stopPrank();
219
220         uint256 currentAllowance = lp.allowance(
221             address(vault),
222             address(router)
223         );
```

```
224         uint256 requiredAllowance = lp.balanceOf(address(vault));
225
226         console.log("Current allowance:", currentAllowance);
227         console.log("Required allowance:", requiredAllowance);
228
229         // If we call redeem(), expectRevert won't be able to pick up
                the revert, because it happens during a nested call
230         // expectRevert checks the top level call for revert
231         // removeLiquidity() and swapExactTokensForTokens() are nested
                calls inside redeem() which is the top call, that's why
                removeLiquidity() is simulated separatelly here
232
233         // Next removeLiquidity() attempt should revert because of
                insufficient allowance
234         vm.expectRevert(
235             abi.encodeWithSelector(
236                 ERC20InsufficientAllowance.selector,
237                 address(router),
238                 currentAllowance,
239                 requiredAllowance
240             )
241         );
242
243         vm.startPrank(address(vault));
244         router.removeLiquidity(
245             address(asset),
246             address(mockWETH),
247             depositAmt,
248             0,
249             0,
250             user,
251             1
252         );
253         vm.stopPrank();
254     }
255 }
```

Recommended Mitigation:

Add safe increase allowance for LP and Counterparty Token

```
1     function _uniswapDivest(IERC20 token, uint256 liquidityAmount)
          internal returns (uint256 amountOfAssetReturned) {
2         IERC20 counterPartyToken = token == i_weth ? i_tokenOne :
              i_weth;
3
4         // Approve uniswapRouter to transfer LP from VaultShares
5 +       if (
6 +           s_uniswapLP.allowance(address(this), address(
      i_uniswapRouter)) == 0
7 +         ) {
```

```
+               s_uniswapLP.safeIncreaseAllowance(
+                   address(i_uniswapRouter),
+                   type(uint256).max
+               );
+           };

        (uint256 tokenAmount, uint256 counterPartyTokenAmount) =
            i_uniswapRouter.removeLiquidity({
            tokenA: address(token),
            tokenB: address(counterPartyToken),
            liquidity: liquidityAmount,
            amountAMin: 0,
            amountBMin: 0,
            to: address(this),
            deadline: block.timestamp
        });
        s_pathArray = [address(counterPartyToken), address(token)];


        // Approve uniswapRouter to transfer counterPartyToken from
            VaultShares
+       if (
+            counterPartyToken.allowance(address(this), address(
    i_uniswapRouter)) == 0
+       ) {
+           counterPartyToken.safeIncreaseAllowance(
+               address(i_uniswapRouter),
+               type(uint256).max
+           );
+           };

        uint256[] memory amounts = i_uniswapRouter.
            swapExactTokensForTokens({
            amountIn: counterPartyTokenAmount,
            amountOutMin: 0,
            path: s_pathArray,
            to: address(this),
            deadline: block.timestamp
        });
        emit UniswapDivested(tokenAmount, amounts[1]);
        amountOfAssetReturned = amounts[1];
    }
```

[H-4] Guardians have no restrictions on minting vgToken and can accumulate high amounts to take over DAO, steal DAO fees and set parameters for own interest

Description: Becoming a guardian is repaied by receiving Vault Guardian Tokens (vgTokens). Whenever a guardian successfully calls `VaultGuardiansBase::becomeGuardian` or

`VaultGuardiansBase::becomeTokenGuardian`, `_becomeTokenGuardian` is executed, which mints the caller `i_vgToken`.

```
    function _becomeTokenGuardian(IERC20 token, VaultShares tokenVault)
        private returns (address) {
        s_guardians[msg.sender][token] = IVaultShares(address(
            tokenVault));
@>      i_vgToken.mint(msg.sender, s_guardianStakePrice);
        emit GuardianAdded(msg.sender, token);
        token.safeTransferFrom(msg.sender, address(this),
            s_guardianStakePrice);
        token.approve(address(tokenVault), s_guardianStakePrice);
        tokenVault.deposit(s_guardianStakePrice, msg.sender);
        return address(tokenVault);
    }
```

Guardians are also free to quit their role at any time, calling the `VaultGuardianBase::quitGuardian` function. The combination of minting vgTokens, and freely being able to quit, results in users being able to farm vgTokens at any time.

Impact: Assuming the token has no monetary value, the malicious guardian could accumulate tokens until they can overtake the DAO. Then, they could execute any of these functions of the `VaultGuardians` contract:

```
  "sweepErc20s(address)": "942d0ff9",
  "transferOwnership(address)": "f2fde38b",
  "updateGuardianAndDaoCut(uint256)": "9e8f72a4",
  "updateGuardianStakePrice(uint256)": "d16fe105",
```

Proof of Concept:

1. User becomes WETH guardian and is minted vgTokens.
2. User quits, is given back original WETH allocation.
3. User becomes WETH guardian with the same initial allocation.
4. Repeat to keep minting vgTokens indefinitely.

Code

Place the following code into `VaultGuardiansBaseTest.t.sol`

```
    function testDaoTakeover() public hasGuardian hasTokenGuardian {
        address maliciousGuardian = makeAddr("maliciousGuardian");
        uint256 startingVoterUsdcBalance = usdc.balanceOf(
            maliciousGuardian);
        uint256 startingVoterWethBalance = weth.balanceOf(
            maliciousGuardian);
        assertEq(startingVoterUsdcBalance, 0);
        assertEq(startingVoterWethBalance, 0);
```

```
7
8           VaultGuardianGovernor governor = VaultGuardianGovernor(payable(
                vaultGuardians.owner()));
9           VaultGuardianToken vgToken = VaultGuardianToken(address(
                governor.token()));
10
11          // Flash loan the tokens, or just buy a bunch for 1 block
12          weth.mint(mintAmount, maliciousGuardian); // The same amount as
                 the other guardians
13          uint256 startingMaliciousVGTokenBalance = vgToken.balanceOf(
                maliciousGuardian);
14          uint256 startingRegularVGTokenBalance = vgToken.balanceOf(
                guardian);
15          console.log("Malicious vgToken Balance:\t",
                startingMaliciousVGTokenBalance);
16          console.log("Regular vgToken Balance:\t",
                startingRegularVGTokenBalance);
17
18          // Malicious Guardian farms tokens
19          vm.startPrank(maliciousGuardian);
20          weth.approve(address(vaultGuardians), type(uint256).max);
21          for (uint256 i; i < 10; i++) {
22              address maliciousWethSharesVault = vaultGuardians.
                    becomeGuardian(allocationData);
23              IERC20(maliciousWethSharesVault).approve(
24                  address(vaultGuardians),
25                  IERC20(maliciousWethSharesVault).balanceOf(
                        maliciousGuardian)
26              );
27              vaultGuardians.quitGuardian();
28          }
29          vm.stopPrank();
30
31          uint256 endingMaliciousVGTokenBalance = vgToken.balanceOf(
                maliciousGuardian);
32          uint256 endingRegularVGTokenBalance = vgToken.balanceOf(
                guardian);
33          console.log("Malicious vgToken Balance:\t",
                endingMaliciousVGTokenBalance);
34          console.log("Regular vgToken Balance:\t",
                endingRegularVGTokenBalance);
35      }
```

Recommended Mitigation:

1. Mint vgTokens on a vesting schedule after a user becomes a guardian.
2. Burn vgTokens when a guardian quits.

[H-5] Double accounting of swap amount leads to excess token spending and fund loss or DoS

Description:

In `UniswapAdapter::_uniswapInvest`, the function uses amountADesired = amountOfTokenToSwap + amounts[0] when adding liquidity. However, amounts[0] already equals amountOfTokenToSwap, which was spent during the prior swap. This results in double accounting for the same amount of the base token.

```
        (
            uint256 tokenAmount,
            uint256 counterPartyTokenAmount,
            uint256 liquidity
        ) = i_uniswapRouter.addLiquidity({
                tokenA: address(token),
                tokenB: address(counterPartyToken),
                // Double accounting for amountOfTokenToSwap,
                    amountOfTokenToSwap + amounts[0]
                amountADesired: amountOfTokenToSwap + amounts[0],
                amountBDesired: amounts[1],
                amountAMin: 0,
                amountBMin: 0,
                to: address(this),
                deadline: block.timestamp
            });
```

Impact:

The vault attempts to transfer 1.5× the intended investment amount.

If balance is insufficient → the call reverts, halting deposits or investments.

If balance exists → the vault overspends user funds into the liquidity pool, causing a permanent loss of assets and inaccurate share accounting.

Proof of Concept:

Case 1. Investing 100 tokens, the vault spends 150 — proving double counting and extra spending/loss of funds.

Case 2. If the vault lacks the extra 50 tokens, the same path reverts due to insufficient balance, resulting in denial of service.

Proof of Code for Case 1:

Add this code in a new file in test/ProofOfCodes:

Code

```
// SPDX-License-Identifier: MIT
```

```
2  pragma solidity ^0.8.20;
3
4  import "forge-std/Test.sol";
5  import {VaultShares} from "../../src/protocol/VaultShares.sol";
6  import {IVaultShares, IVaultData} from "../../src/interfaces/
       IVaultShares.sol";
7  import {ERC20Mock} from "../mocks/ERC20Mock.sol";
8  import {DataTypes} from "../../src/vendor/DataTypes.sol";
9
10 // ---------------------------
11 //        Uniswap mocks
12 // ---------------------------
13
14 contract MaliciousLP is ERC20Mock {
15     // Inherit ERC20Mock: free mint/burn helpers already available
16 }
17
18 contract MaliciousFactory {
19     address public immutable pair;
20
21     constructor(address pair_) {
22         pair = pair_;
23     }
24
25     function getPair(address, address) external view returns (address)
           {
26         return pair;
27     }
28 }
29
30 contract MaliciousRouter {
31     // This router is Uniswap-V2-shaped enough for VaultShares.
           UniswapAdapter
32     // Behavior:
33     //  - swapExactTokensForTokens: takes all input, returns *1 wei* as
           output
34     //  - addLiquidity: pulls all desired tokens, mints *1 wei* LP
35     //  - removeLiquidity: burns all LP, returns *1 wei* of each token
36     // All succeed because min amounts are zero.
37
38     address private _factory;
39     MaliciousLP private _lp;
40
41     constructor(address factory_, address lp_) {
42         _factory = factory_;
43         _lp = MaliciousLP(lp_);
44     }
45
46     function factory() external view returns (address) {
47         return _factory;
48     }
```

```
49
50      function swapExactTokensForTokens(
51          uint256 amountIn,
52          uint256 /* amountOutMin */, // zero in vulnerable code
53          address[] calldata path,
54          address to,
55          uint256 /* deadline */
56      ) external returns (uint256[] memory amounts) {
57          // pull input
58          ERC20Mock(path[0]).transferFrom(msg.sender, address(this),
                amountIn);
59
60          // mint or transfer *1 wei* of output to `to`
61          // use mint to avoid needing pre-funding
62          ERC20Mock(path[1]).mint(1, to);
63
64          amounts = new uint256[](path.length);
65          amounts[0] = amountIn; // uniswap-style return
66          amounts[1] = 1;
67      }
68
69      function addLiquidity(
70          address tokenA,
71          address tokenB,
72          uint256 amountADesired,
73          uint256 amountBDesired,
74          uint256 amountAMin, // zero in vulnerable code
75          uint256 amountBMin, // zero in vulnerable code
76          address to,
77          uint256 /* deadline */
78      ) external returns (uint256 amountA, uint256 amountB, uint256
            liquidity) {
79          // pull all desired tokens from caller
80          ERC20Mock(tokenA).transferFrom(
81              msg.sender,
82              address(this),
83              amountADesired
84          );
85          ERC20Mock(tokenB).transferFrom(
86              msg.sender,
87              address(this),
88              amountBDesired
89          );
90
91          // min checks are zero -> always satisfied
92          require(amountAMin == 0 && amountBMin == 0, "not the vulnerable
                path");
93
94          // mint only 1 wei of LP to the vault (terrible deal)
95          _lp.mint(1, to);
96
```

```
97          return (amountADesired, amountBDesired, 1);
98      }
99
100     function removeLiquidity(
101         address tokenA,
102         address tokenB,
103         uint256 liquidity,
104         uint256 amountAMin, // zero in vulnerable code
105         uint256 amountBMin, // zero in vulnerable code
106         address to,
107         uint256 /* deadline */
108     ) external returns (uint256 amountA, uint256 amountB) {
109         // take the LP from caller
110         _lp.transferFrom(msg.sender, address(this), liquidity);
111         // burn
112         _lp.burn(liquidity, address(this));
113
114         // return *1 wei* of each token to the vault
115         ERC20Mock(tokenA).mint(1, to);
116         ERC20Mock(tokenB).mint(1, to);
117
118         require(amountAMin == 0 && amountBMin == 0, "not the vulnerable
                path");
119
120         return (1, 1);
121     }
122 }
123
124 // ---------------------------
125 //         Aave mocks
126 // ---------------------------
127
128 // Deploy a small MockAavePool that returns (aTokenAddress, 0,0,...)
129 contract MockAavePool {
130     address public immutable aTokenAddr;
131
132     constructor(address _aTokenAddr) {
133         aTokenAddr = _aTokenAddr;
134     }
135
136     // Must match the interface signature used by VaultShares
137     function getReserveData(
138         address
139     ) external view returns (DataTypes.ReserveData memory r) {
140         // leave everything at the zero default, just set aTokenAddress
141         r.aTokenAddress = aTokenAddr;
142     }
143
144     function supply(
145         address /* asset */,
146         uint256 amount,
```

```
147             address onBehalfOf,
148             uint16 /* referralCode */
149         ) external {
150             if (amount > 0) {
151                 ERC20Mock(aTokenAddr).mint(amount, onBehalfOf);
152             }
153         }
154     }
155
156     // ---------------------------
157     //          PoC test
158     // ---------------------------
159
160     contract SlippageZeroMin_PoC_Test is Test {
161         VaultShares private vault;
162         ERC20Mock private asset;
163         ERC20Mock private tokenOne;
164         ERC20Mock private mockWETH; // distinct mock WETH
165         ERC20Mock private awethTokenMock;
166
167         MaliciousLP lp;
168         MaliciousFactory factory;
169         MaliciousRouter router;
170
171         function setUp() public {
172             asset = new ERC20Mock();
173             tokenOne = new ERC20Mock();
174             mockWETH = new ERC20Mock(); // distinct mock WETH
175             awethTokenMock = new ERC20Mock();
176
177             MockAavePool pool = new MockAavePool(address(awethTokenMock));
178
179             // LP + Factory + Router (malicious)
180             lp = new MaliciousLP();
181             factory = new MaliciousFactory(address(lp));
182             router = new MaliciousRouter(address(factory), address(lp));
183
184             // Build constructor data with:
185             IVaultShares.ConstructorData memory c = IVaultShares.
                    ConstructorData({
186                 asset: asset,
187                 vaultName: "PoC Vault",
188                 vaultSymbol: "POC",
189                 guardian: address(this),
190                 allocationData: IVaultData.AllocationData({
191                     holdAllocation: 0,
192                     uniswapAllocation: 1000, // 100% to Uniswap path (will
                            hit vulnerable code)
193                     aaveAllocation: 0
194                 }),
195                 aavePool: address(pool),
```

```
196                 uniswapRouter: address(router),
197                 guardianAndDaoCut: 100,
198                 vaultGuardians: address(this),
199                 weth: address(mockWETH),
200                 usdc: address(tokenOne)
201             });
202
203             vault = new VaultShares(c);
204
205             // Label for nicer traces
206             vm.label(address(vault), "VaultShares(POC)");
207             vm.label(address(asset), "ASSET");
208             vm.label(address(tokenOne), "TOKEN_ONE");
209             vm.label(address(factory), "MaliciousFactory");
210             vm.label(address(router), "MaliciousRouter");
211             vm.label(address(lp), "MaliciousLP");
212         }
213
214         function test_Slippage_WithZeroMins_LosesValue() public {
215             // User mints and deposits 100 ASSET
216             address user = address(0xBEEF);
217             uint256 depositAmt = 100 ether;
218
219             asset.mint(depositAmt, user);
220
221             // pre-fund the vault so the double-accounting won't cause
                    insufficient balance
222             // this must happen after vault creation (setUp) but before
                    deposit triggers invest()
223             asset.mint(depositAmt / 2, address(vault));
224
225             // Approve & deposit
226             vm.startPrank(user);
227             asset.approve(address(vault), depositAmt);
228             uint256 shares = vault.deposit(depositAmt, user);
229             vm.stopPrank();
230
231             // Adding the uniswap approvals that are missing in the
                    original code
232             vm.startPrank(address(vault));
233             lp.approve(address(router), type(uint256).max);
234             tokenOne.approve(address(router), type(uint256).max);
235             mockWETH.approve(address(router), type(uint256).max);
236             vm.stopPrank();
237
238             // Full redeem immediately - vulnerable paths:
239             //  - invest: swap with amountOutMin=0, addLiquidity with
                    amountAMin/BMin=0
240             //  - redeem: removeLiquidity with amountAMin/BMin=0, swap with
                     amountOutMin=0
241             vm.startPrank(user);
```

```
242         uint256 assetsOut = vault.redeem(vault.balanceOf(user), user,
                user);
243         vm.stopPrank();
244
245         // Show the damage in logs
246         console.log("Deposit:", depositAmt);
247         console.log("Shares minted:", shares);
248         console.log("Assets out on full redeem:", assetsOut);
249
250         // Assert a large loss -> 10x
251         assertLt(
252             assetsOut,
253             depositAmt / 10,
254             "Expected significant loss due to zero slippage protections
                "
255         );
256     }
257 }
```

Recommended Mitigation:

Use only the remaining base tokens for amountADesired:

```
1  +        uint256 tokenLeft = amount - amounts[0];
2
3           (
4               uint256 tokenAmount,
5               uint256 counterPartyTokenAmount,
6               uint256 liquidity
7           ) = i_uniswapRouter.addLiquidity({
8                   tokenA: address(token),
9                   tokenB: address(counterPartyToken),
10                  // Use only the remaining base tokens after swap for
                        amountADesired
11 +                amountADesired: uint256 tokenLeft,
12                  amountBDesired: amounts[1],
13                  amountAMin: 0,
14                  amountBMin: 0,
15                  to: address(this),
16                  deadline: block.timestamp
17              });
```

[H-6] No slippage protection in uniswap operations in `UniswapAdapter` can lead to loss of funds due to price changes

Description:

The contract performs Uniswap swaps and liquidity operations (swapExactTokensForTokens, addLiquidity, and removeLiquidity) without specifying minimum acceptable output or input values. This lack

of slippage control allows a transaction to execute at any unfavorable price, exposing the protocol to MEV, frontrunning, and large value loss due to market movement.

- Case 1: 0 value for `amountOutMin` in swaps

Found in:

`UniswapAdapter::_uniswapInvest` -> swapExactTokensForTokens() `UniswapAdapter::_uniswapDivest` -> swapExactTokensForTokens()

```
uint256[] memory amounts = i_uniswapRouter.swapExactTokensForTokens({
    amountIn: amountOfTokenToSwap,
    amountOutMin: 0, //  no protection
    ...
});
```

Setting these to 0 means the transaction will execute even if the received output is 99% below expected. A small market move, a MEV sandwich, or a manipulated pair could drain value from the vault.

For example, when calling `swapExactTokensForTokens()` with amountOutMin = 0, there is no lower limit on the amount of returned tokens, accepting whatever value the market gives back.

Uniswap's price depends on pool reserves, so if the pool moves (price shifts between transaction submission and execution), or a MEV bot front-runs the transaction and manipulates the pool price, a large percentage of tokens can be lost.

- Case 2: 0 value for `amountAMin` / `amountBMin` in liquidity operations

Found in:

`UniswapAdapter::_uniswapInvest` -> addLiquidity() `UniswapAdapter::_uniswapDivest` -> removeLiquidity()

```
(, , uint256 liquidity) = i_uniswapRouter.addLiquidity({
    ...
    amountAMin: 0, //  no protection
    amountBMin: 0, //  no protection
    ...
});

(, ) = i_uniswapRouter.removeLiquidity({
    ...
    amountAMin: 0, //  no protection
    amountBMin: 0, //  no protection
    ...
});
```

When the contract supplies or removes liquidity from a Uniswap pool with amountAMin and amountBMin set to 0, it effectively disables any price-slippage protection during these operations.

Uniswap uses these parameters to guarantee that liquidity is only added or removed within an acceptable price range.

If both are zero, the function call will succeed regardless of how far the pool's price has moved since the transaction was initiated.

Impact: Loss of vault funds -> An attacker or validator can manipulate price before execution, leading to vault assets being swapped at disadvantageous rates, draining value from user deposits.

Proof of Concept:

Exploit scenario: Swap with amountOutMin = 0 & Add liquidity with amountAMin / amountBMin = 0

Goal: Manipulate price so the vault swaps into an unfavorable ratio before addLiquidity, causing the vault to deposit assets at a bad price (lose value / mint fewer LP).

1. A user deposits weth into the vault which triggers _investFunds. Vault will attempt to swap half the deposited token for the counterparty and then call addLiquidity with amountAMin = 0 and amountBMin = 0.
2. Attacker front-run / manipulate price: Immediately before the vault's swap or addLiquidity will be included in a block, the attacker executes a large swap (or flash-loan + swap) on the same pair to move the pool price drastically (e.g., swap many USDC for WETH)
3. Vault executes swap with amountOutMin = 0, because amountOutMin is zero, the router will not revert — the vault accepts the unfavorable output amount.
4. Vault calls addLiquidity(amountAMin=0, amountBMin=0), with the skewed balances, the vault adds liquidity at the manipulated ratio and receives fewer LP tokens than expected or pairs tokens in a bad ratio — effectively losing value.
5. Vault deposits at manipulated price and loses value. No revert occurs because amountOutMin and amountAMin/amountBMin are zero.

- Proof of Code:

Add this code in a new file in test/ProofOfCodes:

Code

```
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.20;

import "forge-std/Test.sol";
import {VaultShares} from "../../src/protocol/VaultShares.sol";
import {IVaultShares, IVaultData} from "../../src/interfaces/
    IVaultShares.sol";
import {ERC20Mock} from "../mocks/ERC20Mock.sol";
import {DataTypes} from "../../src/vendor/DataTypes.sol";

```

```
// ---------------------------
//        Uniswap mocks
// ---------------------------

contract MaliciousLP is ERC20Mock {
    // Inherit ERC20Mock: free mint/burn helpers already available
}

contract MaliciousFactory {
    address public immutable pair;

    constructor(address pair_) {
        pair = pair_;
    }

    function getPair(address, address) external view returns (address) {
        return pair;
    }
}

contract MaliciousRouter {
    // This router is Uniswap-V2-shaped enough for VaultShares.UniswapAdapter
    // Behavior:
    //  - swapExactTokensForTokens: takes all input, returns *1 wei* as output
    //  - addLiquidity: pulls all desired tokens, mints *1 wei* LP
    //  - removeLiquidity: burns all LP, returns *1 wei* of each token
    // All succeed because min amounts are zero.

    address private _factory;
    MaliciousLP private _lp;

    constructor(address factory_, address lp_) {
        _factory = factory_;
        _lp = MaliciousLP(lp_);
    }

    function factory() external view returns (address) {
        return _factory;
    }

    function swapExactTokensForTokens(
        uint256 amountIn,
        uint256 /* amountOutMin */, // zero in vulnerable code
        address[] calldata path,
        address to,
        uint256 /* deadline */
    ) external returns (uint256[] memory amounts) {
        // pull input
```

```
58          ERC20Mock(path[0]).transferFrom(msg.sender, address(this),
                amountIn);
59
60          // mint or transfer *1 wei* of output to `to`
61          // use mint to avoid needing pre-funding
62          ERC20Mock(path[1]).mint(1, to);
63
64          amounts = new uint256[](path.length);
65          amounts[0] = amountIn; // uniswap-style return
66          amounts[1] = 1;
67      }
68
69      function addLiquidity(
70          address tokenA,
71          address tokenB,
72          uint256 amountADesired,
73          uint256 amountBDesired,
74          uint256 amountAMin, // zero in vulnerable code
75          uint256 amountBMin, // zero in vulnerable code
76          address to,
77          uint256 /* deadline */
78      ) external returns (uint256 amountA, uint256 amountB, uint256
            liquidity) {
79          // pull all desired tokens from caller
80          ERC20Mock(tokenA).transferFrom(
81              msg.sender,
82              address(this),
83              amountADesired
84          );
85          ERC20Mock(tokenB).transferFrom(
86              msg.sender,
87              address(this),
88              amountBDesired
89          );
90
91          // min checks are zero -> always satisfied
92          require(amountAMin == 0 && amountBMin == 0, "not the vulnerable
                path");
93
94          // mint only 1 wei of LP to the vault (terrible deal)
95          _lp.mint(1, to);
96
97          return (amountADesired, amountBDesired, 1);
98      }
99
100     function removeLiquidity(
101         address tokenA,
102         address tokenB,
103         uint256 liquidity,
104         uint256 amountAMin, // zero in vulnerable code
105         uint256 amountBMin, // zero in vulnerable code
```

```
106             address to,
107             uint256 /* deadline */
108         ) external returns (uint256 amountA, uint256 amountB) {
109             // take the LP from caller
110             _lp.transferFrom(msg.sender, address(this), liquidity);
111             // burn
112             _lp.burn(liquidity, address(this));
113
114             // return *1 wei* of each token to the vault
115             ERC20Mock(tokenA).mint(1, to);
116             ERC20Mock(tokenB).mint(1, to);
117
118             require(amountAMin == 0 && amountBMin == 0, "not the vulnerable
                    path");
119
120             return (1, 1);
121         }
122     }
123
124     // ---------------------------
125     //         Aave mocks
126     // ---------------------------
127
128     // Deploy a small MockAavePool that returns (aTokenAddress, 0,0,...)
129     contract MockAavePool {
130         address public immutable aTokenAddr;
131
132         constructor(address _aTokenAddr) {
133             aTokenAddr = _aTokenAddr;
134         }
135
136         // Must match the interface signature used by VaultShares
137         function getReserveData(
138             address
139         ) external view returns (DataTypes.ReserveData memory r) {
140             // leave everything at the zero default, just set aTokenAddress
141             r.aTokenAddress = aTokenAddr;
142         }
143
144         function supply(
145             address /* asset */,
146             uint256 amount,
147             address onBehalfOf,
148             uint16 /* referralCode */
149         ) external {
150             if (amount > 0) {
151                 ERC20Mock(aTokenAddr).mint(amount, onBehalfOf);
152             }
153         }
154     }
155
```

```
156 // ---------------------------
157 //          PoC test
158 // ---------------------------
159
160 contract SlippageZeroMin_PoC_Test is Test {
161     VaultShares private vault;
162     ERC20Mock private asset;
163     ERC20Mock private tokenOne;
164     ERC20Mock private mockWETH; // distinct mock WETH
165     ERC20Mock private awethTokenMock;
166
167     function setUp() public {
168         asset = new ERC20Mock();
169         tokenOne = new ERC20Mock();
170         mockWETH = new ERC20Mock(); // distinct mock WETH
171         awethTokenMock = new ERC20Mock();
172
173         MockAavePool pool = new MockAavePool(address(awethTokenMock));
174
175         // LP + Factory + Router (malicious)
176         MaliciousLP lp = new MaliciousLP();
177         MaliciousFactory factory = new MaliciousFactory(address(lp));
178         MaliciousRouter router = new MaliciousRouter(
179             address(factory),
180             address(lp)
181         );
182
183         // Build constructor data with:
184         IVaultShares.ConstructorData memory c = IVaultShares.
               ConstructorData({
185             asset: asset,
186             vaultName: "PoC Vault",
187             vaultSymbol: "POC",
188             guardian: address(this),
189             allocationData: IVaultData.AllocationData({
190                 holdAllocation: 0,
191                 uniswapAllocation: 1000, // 100% to Uniswap path (will
                       hit vulnerable code)
192                 aaveAllocation: 0
193             }),
194             aavePool: address(pool),
195             uniswapRouter: address(router),
196             guardianAndDaoCut: 100,
197             vaultGuardians: address(this),
198             weth: address(mockWETH),
199             usdc: address(tokenOne)
200         });
201
202         vault = new VaultShares(c);
203
204         // Label for nicer traces
```

```
205             vm.label(address(vault), "VaultShares(POC)");
206             vm.label(address(asset), "ASSET");
207             vm.label(address(tokenOne), "TOKEN_ONE");
208             vm.label(address(factory), "MaliciousFactory");
209             vm.label(address(router), "MaliciousRouter");
210             vm.label(address(lp), "MaliciousLP");
211         }
212
213         function test_Slippage_WithZeroMins_LosesValue() public {
214             // User mints and deposits 100 ASSET
215             address user = address(0xBEEF);
216             uint256 depositAmt = 100 ether;
217
218             asset.mint(depositAmt, user);
219
220             // pre-fund the vault so the double-accounting won't cause
                    insufficient balance
221             // this must happen after vault creation (setUp) but before
                    deposit triggers invest()
222             asset.mint(depositAmt / 2, address(vault));
223
224             // Approve & deposit
225             vm.startPrank(user);
226             asset.approve(address(vault), depositAmt);
227             uint256 shares = vault.deposit(depositAmt, user);
228             vm.stopPrank();
229
230             // Full redeem immediately - vulnerable paths:
231             //  - invest: swap with amountOutMin=0, addLiquidity with
                    amountAMin/BMin=0
232             //  - redeem: removeLiquidity with amountAMin/BMin=0, swap with
                     amountOutMin=0
233             vm.startPrank(user);
234             uint256 assetsOut = vault.redeem(vault.balanceOf(user), user,
                    user);
235             vm.stopPrank();
236
237             // Show the damage in logs
238             console.log("Deposit:", depositAmt);
239             console.log("Shares minted:", shares);
240             console.log("Assets out on full redeem:", assetsOut);
241
242             // Assert a large loss -> 10x
243             assertLt(
244                 assetsOut,
245                 depositAmt / 10,
246                 "Expected significant loss due to zero slippage protections
                        "
247             );
248         }
249     }
```

Recommended Mitigation:

Set a real slippage tolerance -> Allow the user (or the vault logic) to specify a percentage tolerance for price movement.

- Case 1:

Compute `amountOutMin` dynamically before swapping.

Add a new function to dynamically calculate `amountOutMin` based on user entered slippage tolerance `slippageBps`:

```
+    function _getAmountOutMin(
+        IUniswapV2Router01 router,
+        address[] memory path,
+        uint256 amountIn,
+        uint256 slippageBps // e.g., 100 = 1%
+    ) internal view returns (uint256) {
+        uint256[] memory amountsOut = router.getAmountsOut(amountIn, path);
+        uint256 expectedOut = amountsOut[amountsOut.length - 1];
+        return (expectedOut * (10_000 - slippageBps)) / 10_000;
+    }
```

Use this value to assign it to `amountOutMinin` for `swapExactTokensForTokens` in `_uniswapInvest` and `uniswapDivest`:

```
+    function _uniswapInvest(IERC20 token, uint256 amount, uint256 slippageBps) internal {
         IERC20 counterPartyToken = token == i_weth ? i_tokenOne : i_weth;
         uint256 amountOfTokenToSwap = amount / 2;
         s_pathArray = [address(token), address(counterPartyToken)];

         bool succ = token.approve(address(i_uniswapRouter), amountOfTokenToSwap);
         if (!succ) {
             revert UniswapAdapter__TransferFailed();
         }

         // Compute slippage-protected min output
+        uint256 minOut = _getAmountOutMin(i_uniswapRouter, s_pathArray, half, slippageBps);

         uint256[] memory amounts = i_uniswapRouter.
             swapExactTokensForTokens({
             amountIn: amountOfTokenToSwap,
+            amountOutMin: minOut,
             path: s_pathArray,
             to: address(this),
```

```
            deadline: block.timestamp
        });

        ...
    }
```

- Case 2:

Add a new function to dynamically calculate `amountAMin` / `amountBMin` based on user entered slippage tolerance `slippageBps`, like in the swaps example:

```
+    function _calculateLiquidityMin(
+        uint256 amountADesired,
+        uint256 amountBDesired,
+        uint256 slippageBps // e.g., 100 = 1%
+    ) internal pure returns (uint256 amountAMin, uint256 amountBMin) {
+        amountAMin = (amountADesired * (10_000 - slippageBps)) / 10_000;
+        amountBMin = (amountBDesired * (10_000 - slippageBps)) / 10_000;
+    }
```

Use this values to assign them to `amountAMin` / `amountBMin` for `addLiquidity`/`removeLiquidity` in `_uniswapInvest` and `uniswapDivest`:

```
+    (uint256 tokenAMin, uint256 tokenBMin) = _calculateLiquidityMin(
+        desiredA,
+        desiredB,
+        100 // 1% slippage tolerance
+    );

     (
        uint256 tokenAmount,
        uint256 counterPartyTokenAmount,
        uint256 liquidity
     ) = i_uniswapRouter.addLiquidity({
        tokenA: address(token),
        tokenB: address(counterPartyToken),
        amountADesired: desiredA,
        amountBDesired: desiredB,
+       amountAMin: tokenAMin,
+       amountBMin: tokenBMin,
        to: address(this),
        deadline: block.timestamp
     });
```

Same for `removeLiquidity`.

[H-7] Uniswap LP token can be `address(0)` when vault asset is `weth`(weth/weth pair), causing functional failure and potential token loss.

Description: In `VaultShares` in constructor, is fetched the uniswap LP token.

```
i_uniswapLiquidityToken = IERC20(
    i_uniswapFactory.getPair(address(token), address(counterPartyToken)
        )
);
```

There was no validation that getPair() actually returned a valid deployed pair. If the vault's token and counterPartyToken are identical (e.g., both WETH), Uniswap V2's getPair() returns address(0) since no such pool can exist.

This means the vault would set:

```
i_uniswapLiquidityToken = IERC20(address(0));
```

and continue execution without reverting. Later, when the vault attempts to call `i_uniswapLiquidityToken.balanceOf()`, or Transfer LP tokens, or Redeem liquidity, the transaction reverts with:

```
call to non-contract address 0x0000000000000000000000000000000000000000
```

Now the vault becomes permanently unusable (funds could be stuck).

Impact:

Vault logic dependent on LP token calls fails completely. Vault cannot invest, withdraw, or divest. Assets may become stuck or lost if the vault's state becomes unrecoverable.

The issue breaks core functionality and potentially leads to permanent asset loss or vault lockup.

Proof of Concept:

Reproduction:

1. Vault initializes with WETH as asset.
2. `_uniswapInvest()` calls getPair(WETH, WETH) → returns address(0).
3. Vault sets i_uniswapLiquidityToken = IERC20(address(0)).
4. On withdrawal or reinvestment, vault reverts with: "call to non-contract address 0x0000000000000000000000000000000000000000
5. Vault becomes permanently stuck.

- Proof of Code:

Add this test in WethFork.t.sol in test/fork

Mock Uniswap Factory getPair() always returns a valid address, unlike real mainnet or forked scenario where address(0) is returned.

Code

```
function testWethVaultGeneratesUniswapAddress0LP() public hasGuardian {
        vm.startPrank(user);
        wETH.deposit{value: mintAmount}(); // convert ETH -> WETH
        wETH.approve(address(wethVaultShares), mintAmount);
        wethVaultShares.deposit(mintAmount, user);

        uniswapLiquidityToken = wethVaultShares.i_uniswapLiquidityToken
            ();

        console.log(
            "Uniswap LP token address:",
            address(uniswapLiquidityToken)
        );

        assertEq(address(uniswapLiquidityToken), address(0));
    }
```

Recommended Mitigation:

The current design only allows weth as counter pair token.

Proposed solution:

1. Add a dynamic detection for uniswap counter pair:
    - asset: weth -> counter pair: usdc
    - asset: not weth -> counter pair: weth

```
      // Dynamic LP counter-asset selection
+     if (constructorData.asset == constructorData.weth) {
          // asset is WETH, so use USDC as the counter asset for Uniswap
              LP
+         i_uniswapPairToken = IERC20(constructorData.usdc);
+     } else {
          // otherwise, standard case - use WETH
+         i_uniswapPairToken = IERC20(constructorData.weth);
+     }

      i_uniswapLiquidityToken = IERC20(
          i_uniswapFactory.getPair(
          address(constructorData.asset),
          // *** use a variable instead of hardcoding i_weth ***
+         address(i_uniswapPairToken)
          )
      );
```

Other solutions, but less ideal:

2. Swap weth to usdc -> adds more complexity and moving parts

3. Redistribution of uniswap allocation into aave allocation -> breaks investment diversification
4. Prohibit weth vaults -> restrictive to users

[H-8] `nonReentrant` is not the First Modifier in `VaultShares` which makes the functions vulnerable to reentrancy attacks

Description:

Found in: `VaultShares::deposit` `VaultShares::rebalanceFunds` `VaultShares::withdraw` `VaultShares::redeem`

The placement of the divestThenInvest modifier before the nonReentrant modifier in the function signature creates a critical window for exploitation.

```
function rebalanceFunds() public isActive divestThenInvest nonReentrant
    {}
```

Impact: Loss or manipulation of funds

This could allow an attacker to withdraw their share twice or call the divestment logic to retrieve tokens that they shouldn't have access to, leading to a direct loss of user funds from the vault.

A reentrant call could interrupt the process, leading to a situation where the contract's internal accounting is calculated based on incomplete or manipulated data.

Proof of Concept:

- Mechanism of Attack

The Solidity compiler executes modifiers in the order they are listed.

1. isActive runs (a simple check, no external call, safe).
2. divestThenInvest runs:

- It executes the Divest logic, which includes external calls to third-party protocols (Uniswap and Aave) via _uniswapDivest and _aaveDivest.
- Since divestment from these protocols often involves the target protocol sending tokens back to the vault, these are external calls that transfer control flow.

3. Reentrancy Window Opens: An attacker (who controls one of the addresses or a token hook involved in the divestment) can trigger a malicious fallback or hook function in their contract during the divestment phase.

4. Bypass: The attacker's malicious code executes and immediately calls rebalanceFunds() again. Since the execution hasn't reached the nonReentrant modifier yet, the reentrancy lock has not been set.
5. Exploitation: The recursive call to rebalanceFunds() proceeds and executes the divestThenInvest logic a second time, potentially causing unauthorized state changes, double-claiming of funds, or manipulation of the fund's internal accounting before the first call can complete its state updates.

Recommended Mitigation:

Always place the nonReentrant modifier first in the function declaration.

```
function rebalanceFunds() public nonReentrant isActive divestThenInvest
    {}
```

[H-9] Using `block.timestamp` for swap deadline offers no protection

Description:

Found in:

`UniswapAdapter::_uniswapInvest` -> swapExactTokensForTokens(), addLiquidity()
`UniswapAdapter::_uniswapDivest` -> swapExactTokensForTokens(), removeLiquidity()

The purpose of the deadline parameter in a swap function (like on Uniswap) is to specify a fixed future time after which the transaction will fail.

When deadline: block.timestamp is used, the deadline is set to the exact moment the transaction is included in the block.

The swap router contract usually checks:

```
require(deadline >= block.timestamp, 'EXPIRED');
```

Impact: Loss of funds(MEV Attacks), transactions stuck forever

The primary function of the swap deadline is to protect the user from slippage and Miner Extractable Value (MEV) attacks, specifically the sandwich attack. By nullifying the deadline check:

Enables Sandwich Attacks: A block producer (validator) can see the user's transaction in the mempool and hold it indefinitely. They can wait for the price to move favorably, execute a transaction just before the user's swap (front-running) to drive the price against the user, and a transaction just after (back-running) to pocket the difference. The user is guaranteed a bad price, and the block producer extracts maximum value.

Allows Arbitrary Delay/Censorship: The validator can delay the transaction for hours or even days, allowing huge price swings that exceed the user's intended slippage tolerance, leading to massive and unintended loss of funds.

The direct result of this vulnerability is the theft or extraction of value from the user, which places it firmly in the high-severity category.

Proof of Concept: If the user passes a fixed future time (e.g., block.timestamp + 300 seconds from the frontend), the check prevents the transaction from being executed after that time.

If the user passes block.timestamp from within the smart contract, the condition becomes require(block.timestamp >= block.timestamp), which will always be true.

The deadline check is completely bypassed, meaning the transaction never expires due to time and offers zero protection against high slippage or attacks by malicious block producers.

Recommended Mitigation:

1. Use a future deadline window Instead of:

```
deadline: block.timestamp
```

use something like:

```
deadline: block.timestamp + 300  // 5 minutes
```

or make it configurable.

That way: Your transaction won't be stuck forever (it expires in 5 minutes). A validator can't hold it indefinitely for manipulation.

2. Use MEV-resistant submission For high-value transactions: Send the transaction via Flashbots Protect / MEV-Blocker RPC or similar RPC relayers. These systems submit your transaction directly to block builders privately (not via the public mempool), preventing front-running and sandwiching.

Medium

[M-1] Using unsafe ERC20 approve operation can result in stolen assets

Description: The ERC20 approve() function sets the spender's allowance to a specific amount — but if there was already an allowance, it overwrites it. This creates a race condition known as the ERC20 approve front-running issue: If someone can front-run your transaction between two approvals (e.g., approve(100) → approve(200)), they might use the old allowance before it's updated.

Used in:

`UniswapAdapter::_uniswapInvestAaveAdapter::_aaveInvestVaultGuardiansBase::_becomeTokenGuardian`

Impact: Loss of funds

Proof of Concept:

You currently have allowance = 100. You try to change it to 200. Before your new approval is mined, the spender calls transferFrom(..., 100) — using the old allowance. Your transaction sets allowance to 200 again → they can now pull 200 more. This risk exists any time you use approve() to modify a nonzero allowance directly.

Recommended Mitigation:

Option 1 — Use SafeERC20.safeIncreaseAllowance OpenZeppelin's SafeERC20 has helper functions that safely add to the existing allowance instead of overwriting it: asset.safeIncreaseAllowance(address(i_aavePool), amount); This increases the allowance by amount rather than replacing it, and reverts on failure. You don't even need to check a boolean return — the SafeERC20 wrapper handles that internally.

Option 2 — Use safeApprove(0) before setting new allowance If you prefer the raw approve() semantics, the safe pattern is: asset.safeApprove(address(i_aavePool), 0); asset.safeApprove(address(i_aavePool), amount); This clears the previous allowance first (per ERC20 recommendations), then sets the new one. It ensures no residual approvals exist in between.

In this specific context (Aave and Uniswap), the spender is a trusted protocol — it will only pull tokens during your supply()/addLiquidity() call, so the race condition isn't a real-world exploit risk.

However, it's still a bad pattern to replicate elsewhere, for example VaultGuardiansBase.sol. If your vault ever interacts with other contracts dynamically (like different pools or adapters), it becomes dangerous. So it's still best practice to always use SafeERC20 helpers for approvals.

[M-2] Centralization Risk

Description: These functions are using an access control library that put the power to make changes in the hands of a single entity, which is a centralization issue.

5 Found Instances

- Found in src/dao/VaultGuardianToken.sol Line: 9

```
1  contract VaultGuardianToken is ERC20, ERC20Permit, ERC20Votes,
       Ownable {
```

- Found in src/dao/VaultGuardianToken.sol Line: 21

```
1     function mint(address to, uint256 amount) external onlyOwner {
```

- Found in src/protocol/VaultGuardians.sol Line: 40

```
1 contract VaultGuardians is Ownable, VaultGuardiansBase {
```

- Found in src/protocol/VaultGuardians.sol Line: 71

```
1     function updateGuardianStakePrice(uint256 newStakePrice)
          external onlyOwner {
```

- Found in src/protocol/VaultGuardians.sol Line: 82

```
1     function updateGuardianAndDaoCut(uint256 newCut) external
          onlyOwner {
```

Impact: Contracts have owners with privileged rights to perform admin tasks and need to be trusted to not perform malicious updates or drain funds.

Recommended Mitigation: Consider some of the following recommandations to avoid centralization models: 1. Use a Multi-signature Wallet. 2. Time-lock Contract. 3. Governance by DAO. 4. Use of AccessControl library from OpenZeppelin, which restricts the access to a group of addresses instead a single one.

[M-3] Missing validation for address(0) in `VaultShares` constructor

Description: In `VaultShares` constructor `aToken` and `Uniswap` pair are assigned by making external calls to aave getReserveData, respectively uniswap factory.

- Aave: aTokenAddress == address(0)

If getReserveData() returns an empty struct because that asset isn't supported by Aave, then aTokenAddress will be address(0).

Consequence: Later, when your vault tries to deposit to Aave via _aaveInvest(), it will interact with address(0), causing a revert (or worse, if unchecked, a failed call without revert detection).

Impact: Users won't lose funds, but deposits/rebalances will revert unexpectedly or result in locked vaults.

- Uniswap: getPair() == address(0)

If the Uniswap factory doesn't have a pair for asset/WETH (e.g., asset == WETH), it will return address(0).

Consequence: _uniswapInvest() and _uniswapDivest() may call the router or LP token assuming a valid pair exists. If not handled, that can cause: reverts during deposits or withdrawals, or invalid assumptions about invested balances.

Impact: No funds can be drained (since address(0) has no code), but the vault may get stuck, or deposits may fail silently.

```
i_aaveAToken = IERC20(
    IPool(constructorData.aavePool)
        .getReserveData(address(constructorData.asset))
        .aTokenAddress
);

i_uniswapLiquidityToken = IERC20(
    i_uniswapFactory.getPair(
        address(constructorData.asset),
        address(i_weth)
    )
);
```

Impact: Vaults can be deployed with unsupported or invalid investment configurations, leading to stuck funds or failed deposits.

Proof of Concept: This proves that a vault can be deployed with `i_aaveAToken` or `i_uniswapLiquidityToken` = address(0)

- Proof of Code:

Add this test in test/ProofOfCodes

Code

```
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.20;

import "forge-std/Test.sol";
import {VaultShares} from "../../src/protocol/VaultShares.sol";
import {IVaultShares, IERC4626, IVaultData} from "../../src/interfaces/
    IVaultShares.sol";
import {DataTypes} from "../../src/vendor/DataTypes.sol";
import {IERC20} from "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import {Base_Test} from "../Base.t.sol";
import {ERC20Mock} from "../mocks/ERC20Mock.sol";

// Mocks returning address(0)
contract MockAavePool is Base_Test {
    function getReserveData(
        address
    ) external pure returns (DataTypes.ReserveData memory) {
        // Create a blank configuration map
```

```
18          DataTypes.ReserveConfigurationMap memory emptyConfig =
                DataTypes
19              .ReserveConfigurationMap({data: 0});
20
21          return
22              DataTypes.ReserveData({
23                  configuration: emptyConfig,
24                  liquidityIndex: 0,
25                  currentLiquidityRate: 0,
26                  variableBorrowIndex: 0,
27                  currentVariableBorrowRate: 0,
28                  currentStableBorrowRate: 0,
29                  lastUpdateTimestamp: 0,
30                  id: 0,
31                  aTokenAddress: address(0),
32                  stableDebtTokenAddress: address(0),
33                  variableDebtTokenAddress: address(0),
34                  interestRateStrategyAddress: address(0),
35                  accruedToTreasury: 0,
36                  unbacked: 0,
37                  isolationModeTotalDebt: 0
38              });
39      }
40  }
41
42  contract MockUniswapFactory {
43      function getPair(address, address) external pure returns (address)
            {
44          return address(0);
45      }
46  }
47
48  contract MockUniswapRouterZero {
49      address private _factory;
50
51      constructor(address factory_) {
52          _factory = factory_;
53      }
54
55      function factory() external view returns (address) {
56          return _factory;
57      }
58  }
59
60  contract VaultSharesZeroAddressPoC is Base_Test {
61      function testDeployVaultWithZeroAddresses() public {
62          // Deploy mocks used for this PoC (independent of Base_Test's
                network config)
63          MockAavePool poolZero = new MockAavePool();
64          MockUniswapFactory factoryZero = new MockUniswapFactory();
65          MockUniswapRouterZero routerZero = new MockUniswapRouterZero(
```

```
66              address(factoryZero)
67          );
68
69          // Deploy simple ERC20 mocks for asset/weth/usdc
70          ERC20Mock mockAsset = new ERC20Mock();
71          ERC20Mock mockWETH = new ERC20Mock();
72          ERC20Mock mockUSDC = new ERC20Mock();
73
74          // Make sure the deployer (this) is the vaultGuardians to
                satisfy onlyVaultGuardians in constructor
75          IVaultShares.ConstructorData memory cdata = IVaultShares
76              .ConstructorData({
77                  asset: mockAsset,
78                  vaultName: "ZeroAddressVault",
79                  vaultSymbol: "ZAV",
80                  aavePool: address(poolZero),
81                  uniswapRouter: address(routerZero),
82                  weth: address(mockWETH),
83                  usdc: address(mockUSDC),
84                  guardian: address(this),
85                  guardianAndDaoCut: 100, // arbitrary
86                  vaultGuardians: address(this), // important:
                        constructor calls updateHoldingAllocation() which is
                         onlyVaultGuardians
87                  allocationData: IVaultData.AllocationData({
88                      holdAllocation: 500,
89                      uniswapAllocation: 250,
90                      aaveAllocation: 250
91                  })
92              });
93
94          // Deploy VaultShares (constructor will query aavePoolZero and
                factoryZero)
95          VaultShares vault = new VaultShares(cdata);
96
97          console.log("Vault deployed at:", address(vault));
98
99          // Vulnerable state: the constructor saved zero addresses
                returned by external calls
100         address aToken = vault.getAaveAToken();
101         address lpToken = vault.getUniswapLiquidtyToken();
102
103         console.log("aToken stored:", aToken);
104         console.log("uniswap LP token stored:", lpToken);
105
106         assertEq(aToken, address(0), "expected aToken to be address(0)"
                );
107         assertEq(
108             lpToken,
109             address(0),
110             "expected uniswap LP token to be address(0)"
```

```
111             );
112         }
113     }
```

Recommended Mitigation: Validate addresses and revert if configuration is invalid.

```
constructor(
        ConstructorData memory constructorData
    )
        ERC4626(constructorData.asset)
        ERC20(constructorData.vaultName, constructorData.vaultSymbol)
        AaveAdapter(constructorData.aavePool)
        UniswapAdapter(
            constructorData.uniswapRouter,
            constructorData.weth,
            constructorData.usdc
        )
    {
        i_guardian = constructorData.guardian;
        i_guardianAndDaoCut = constructorData.guardianAndDaoCut;
        i_vaultGuardians = constructorData.vaultGuardians;
        s_isActive = true;
        updateHoldingAllocation(constructorData.allocationData);

        i_aaveAToken = IERC20(
            IPool(constructorData.aavePool)
                .getReserveData(address(constructorData.asset))
                .aTokenAddress
        );

        // New check for address(0)
+       if (address(i_aaveAToken) == address(0)) {
+           revert("Invalid Aave aToken: unsupported asset");
+       }

        i_uniswapLiquidityToken = IERC20(
            i_uniswapFactory.getPair(
                address(constructorData.asset),
                address(i_weth)
            )
        );

        // New check for address(0)
+       if (address(i_uniswapLiquidityToken) == address(0)) {
+           revert("Uniswap pair not found");
+       }

        i_uniswapLiquidityToken = IERC20(i_uniswapLiquidityToken);
    }
```

[M-4] Anyone can force rebalances causing value leakage and fund instability

Description:

```
function rebalanceFunds() public isActive divestThenInvest nonReentrant
    {}
```

The rebalanceFunds() function is declared as public and can be called by any address, because access is not restricted, for example, onl to guardians. This function fully divests and re-invests all vault assets through external protocol interactions (Uniswap and Aave).

Because it lacks access control, anyone can repeatedly trigger full rebalance cycles. Each rebalance incurs swap fees, gas costs, and potential slippage, negatively affecting vault performance.

Also, this function is already affected by the separate high-severity issue [H-8] “nonReentrant is not the first modifier”, which allows reentrancy during the divestThenInvest phase — amplifying the impact of this unrestricted entrypoint.

Impact:

Attackers or bots can repeatedly call rebalanceFunds() and cause cumulative losses.

Recommended Mitigation:

Restrict access using onlyGuardian or a designated keeper role. Move nonReentrant to be the first modifier, in order to guard against reentrancy.

```
function rebalanceFunds() public nonReentrant onlyGuardian isActive
   divestThenInvest  {}
```

[M-5] Potentially incorrect voting period and delay in governor may affect governance

Description: The `VaultGuardianGovernor` contract, based on OpenZeppelin Contract’s Governor, implements two functions to define the voting delay (`votingDelay`) and period (`votingPeriod`). The contract intends to define a voting delay of 1 day, and a voting period of 7 days. It does it by returning the value `1 days` from `votingDelay` and `7 days` from `votingPeriod`. In Solidity these values are translated to number of seconds.

However, the `votingPeriod` and `votingDelay` functions, by default, are expected to return number of blocks. Not the number seconds.

Impact: Voting period and delay will be far off what the developers intended, which could potentially affect the intended governance mechanics.

Recommended Mitigation:

Consider updating the functions as follows:

```
function votingDelay() public pure override returns (uint256) {
-   return 1 days;
+   return 7200; // 1 day
}

function votingPeriod() public pure override returns (uint256) {
-   return 7 days;
+   return 50400; // 1 week
}
```

[M-6] Low quorum threshold can lead to governance takeover

Description:

The VaultGuardianGovernor contract sets the quorum requirement to only 4% of total votes:

```
GovernorVotesQuorumFraction(4);
```

This means that any proposal reaching just 4% participation is considered valid. In scenarios where token distribution is uneven or voter participation is low, a small minority (or a single large holder) could unilaterally pass proposals, including malicious ones.

Impact: Can affect protocol safety. A 4% quorum threshold makes it feasible for a small coalition or single actor to pass governance actions that alter protocol-critical parameters.

Proof of Concept:

If a tokenholder controls 5% of the governance token supply, they can:

1. Submit a proposal (if proposal threshold allows it).
2. Vote it through with minimal participation.
3. Execute it, gaining control over protocol behavior.

Recommended Mitigation:

Increase quorum fraction to at least 10–20% depending on token distribution and participation expectations.

Introduce a proposal threshold to prevent low-stake or spam proposals.

Low

[L-1] Unused event in `VaultGuardians`

Description:

```
event VaultGuardians__UpdatedFee(uint256 oldFee, uint256 newFee);
```

Impact: Can increase contract size and complexity.

Recommended Mitigation: Consider using or removing the unused event.

[L-2] Unused State Variable in `VaultGuardiansBase`

Description: Declared variable is not used anywhere in the contract logic

```
        uint256 private constant GUARDIAN_FEE = 0.1 ether;
```

Recommended Mitigation: Consider removing it.

[L-3] Unused error in `VaultGuardians`, `VaultGuardiansBase`

Description:

4 Found Instances

- Found in src/protocol/VaultGuardians.sol Line: 43

```
    error VaultGuardians__TransferFailed();
```

- Found in src/protocol/VaultGuardiansBase.sol Line: 46

```
    error VaultGuardiansBase__NotEnoughWeth(uint256 amount,
        uint256 amountNeeded);
```

- Found in src/protocol/VaultGuardiansBase.sol Line: 48

```
    error VaultGuardiansBase__CantQuitGuardianWithNonWethVaults(
        address guardianAddress);
```

- Found in src/protocol/VaultGuardiansBase.sol Line: 51

```
    error VaultGuardiansBase__FeeTooSmall(uint256 fee, uint256
        requiredFee);
```

Impact: Can increase contract size and complexity.

Recommended Mitigation: Consider using or removing the unused errors.

[L-4] PUSH0 Opcode

Description: The primary concern identified in the smart contracts relates to the Solidity compiler version used, specifically pragma solidity 0.8.20;. This version, along with every version after 0.8.19, introduces the use of the PUSH0 opcode. This opcode is not universally supported across all Ethereum Virtual Machine (EVM)-based Layer 2 (L2) solutions. For instance, ZKSync, one of the targeted platforms for this protocol's deployment, does not currently support the PUSH0 opcode.

The consequence of this incompatibility is that contracts compiled with Solidity versions higher than 0.8.19 may not function correctly or fail to deploy on certain L2 solutions.

Impact: The impact of using a Solidity compiler version that includes the PUSH0 opcode is significant for a protocol intended to operate across multiple EVM-based chains. Chains that do not support this opcode will not be able to execute the contracts as intended, resulting in a range of issues from minor malfunctions to complete deployment failures. This limitation directly affects the protocol's goal of wide compatibility and interoperability, potentially excluding it from deployment on key L2 solutions like ZKsync.

Recommended Mitigation: To mitigate this issue and ensure broader compatibility with various EVM-based L2 solutions, it is recommended to downgrade the Solidity compiler version used in the smart contracts to 0.8.19. This version does not utilize the PUSH0 opcode and therefore maintains compatibility with a wider range of L2 solutions, including ZKsync.

[L-5] No address(0) check in VaultGuardiansBase _becomeTokenGuardian for tokenVault and token

Description: These values are passed as parameters, generated by the calling functions (becomeGuardian and becomeTokenGuardian).

The functions (becomeGuardian and becomeTokenGuardian) are actually calling _becomeTokenGuardian and generate a non-zero address via new VaultShares(…) and user input for token. So here the contract logic and the **`private`** visibility help in guarding against assigning an address(0), but there are some edge cases when this could happen, like if the contract is upgraded and new functions or new logic is added.

```
1 s_guardians[msg.sender][token] = IVaultShares(address(tokenVault));
```

Impact: The primary impact of the missing address(0) check in _becomeTokenGuardian is a potential for Corrupted State leading to a localized Denial of Service (DoS), which is only possible if the contract is upgraded, or if another internal function is added that fails to validate the input.

Recommended Mitigation: Add address(0) check before using the values in the `s_guardian` mapping entry.

```
function _becomeTokenGuardian(
        IERC20 token,
        VaultShares tokenVault
    ) private returns (address) {
+       if((address(token) == address(0) {
+          revert VaultGuardiansBase__InvalidToken();
+       if(IVaultShares(address(tokenVault) == address(0) {
+          revert VaultGuardiansBase__InvalidAddress();

        s_guardians[msg.sender][token] = IVaultShares(address(
            tokenVault));
        })
        //...
    }
```

[L-6] The guardian receives its vgToken before the vault creation is confirmed introducing a risk of reentrancy

Description: In order to become a vault guardian a user has to deposit a stake amount `s_guardianStakePrice`. In exchange for this stake amount the vault guardian receives the same amount of `vgToken`, which they use to get their stake amount back when they quit being a vault guardian.

Here, in `VaultGuardiansBase::_becomeTokenGuardian`, the code doesn't follow the Check-Effect-Interaction(CEI) principle and the vault guardian receives the `vgToken` before the tokenVault creation is confirmed successful.

```
function _becomeTokenGuardian(
        IERC20 token,
        VaultShares tokenVault
    ) private returns (address) {
        s_guardians[msg.sender][token] = IVaultShares(address(
            tokenVault));

        emit GuardianAdded(msg.sender, token);

        i_vgToken.mint(msg.sender, s_guardianStakePrice);

        token.safeTransferFrom(msg.sender, address(this),
            s_guardianStakePrice);
```

```
12
13          bool succ = token.approve(address(tokenVault),
                s_guardianStakePrice);
14          if (!succ) {
15              revert VaultGuardiansBase__TransferFailed();
16          }
17          uint256 shares = tokenVault.deposit(s_guardianStakePrice, msg.
                sender);
18          if (shares == 0) {
19              revert VaultGuardiansBase__TransferFailed();
20          }
21          return address(tokenVault);
22      }
```

Impact: Potential loss of funds

Because the function doesn't follow CEI, it could pose a reentrancy risk, but the function is called internally by `VaultGuardiansBase` and the `vgToken` does not include a reentrancy vector, the Early Reward Minting vulnerability is mitigated for the purpose of a reentrancy attack.

Therefore its impact is reduced from a Critical reentrancy to a Low vulnerability.

However it is still a logic flaw (reward -> deposit -> check) and not good practice in function design.

Recommended Mitigation: Consider following the CEI pattern for operations flow. The `vgToken` mint should be moved after the `deposit` check was performed.

```
1  function _becomeTokenGuardian(
2          IERC20 token,
3          VaultShares tokenVault
4      ) private returns (address) {
5          // Checks
6
7          // Effects
8          s_guardians[msg.sender][token] = IVaultShares(address(
               tokenVault));
9
10         emit GuardianAdded(msg.sender, token);
11
12
13         // Interactions
14 -       i_vgToken.mint(msg.sender, s_guardianStakePrice);
15
16         token.safeTransferFrom(msg.sender, address(this),
               s_guardianStakePrice);
17
18         bool succ = token.approve(address(tokenVault),
               s_guardianStakePrice);
19         if (!succ) {
20             revert VaultGuardiansBase__TransferFailed();
21         }
```

```
22          uint256 shares = tokenVault.deposit(s_guardianStakePrice, msg.
                sender);
23          if (shares == 0) {
24              revert VaultGuardiansBase__TransferFailed();
25          }
26          // Correctly executed after checking if deposit was successful
27  +       i_vgToken.mint(msg.sender, s_guardianStakePrice);
28
29          return address(tokenVault);
30      }
```

[L-7] A GuardianAdded event is emitted before the vault creation is confirmed leading to a false positive in the off-chain information flow

Description: In order to become a vault guardian a user has to deposit a stake amount `s_guardianStakePrice`. The event `GuardianAdded` is emitted before checking if the deposit was actually successful. This can lead to inconsistent information for the event readers.

Here, in `VaultGuardiansBase::_becomeTokenGuardian`, the code doesn't follow the Check-Effect-Interaction(CEI) principle and the event is emitted before the state actually changes.

```
1  function _becomeTokenGuardian(
2          IERC20 token,
3          VaultShares tokenVault
4      ) private returns (address) {
5          s_guardians[msg.sender][token] = IVaultShares(address(
               tokenVault));
6
7          emit GuardianAdded(msg.sender, token);
8
9          i_vgToken.mint(msg.sender, s_guardianStakePrice);
10
11         token.safeTransferFrom(msg.sender, address(this),
               s_guardianStakePrice);
12
13         bool succ = token.approve(address(tokenVault),
               s_guardianStakePrice);
14         if (!succ) {
15             revert VaultGuardiansBase__TransferFailed();
16         }
17         uint256 shares = tokenVault.deposit(s_guardianStakePrice, msg.
               sender);
18         if (shares == 0) {
19             revert VaultGuardiansBase__TransferFailed();
20         }
21         return address(tokenVault);
22     }
```

Impact: While the state on-chain is correctly reverted, the misinformation can negatively affect systems relying on logs: front-ends, analytics and other smart contracts.

Proof of Concept:

1. The event is emitted, signaling to the outside world (off-chain indexers, UI dashboards, explorers) that msg.sender has successfully become a guardian.
2. The transaction continues to the deposit call, which fails (e.g., due to insufficient allowance, a vault error, or a low-level call failure).
3. The transaction executes the revert statement, undoing all state changes (including the s_guardians mapping update and the `vgToken` mint).
4. Result: External listeners will see an event log for GuardianAdded in a transaction that ultimately failed and reverted.

This creates a confusing and incorrect picture of the contract's state.

Recommended Mitigation: Consider following the CEI pattern for operations flow. The event emission must be moved to the final section of the function, after all checks and interactions have been successfully completed.

```
function _becomeTokenGuardian(
        IERC20 token,
        VaultShares tokenVault
    ) private returns (address) {
        // Checks

        // Effects
        s_guardians[msg.sender][token] = IVaultShares(address(
            tokenVault));

-       emit GuardianAdded(msg.sender, token);


        // Interactions
-       i_vgToken.mint(msg.sender, s_guardianStakePrice);

        token.safeTransferFrom(msg.sender, address(this),
            s_guardianStakePrice);

        bool succ = token.approve(address(tokenVault),
            s_guardianStakePrice);
        if (!succ) {
            revert VaultGuardiansBase__TransferFailed();
        }
        uint256 shares = tokenVault.deposit(s_guardianStakePrice, msg.
            sender);
        if (shares == 0) {
            revert VaultGuardiansBase__TransferFailed();
```

```
25          }
26          // Correctly executed after checking if deposit was successful
27  +       i_vgToken.mint(msg.sender, s_guardianStakePrice);
28          // Correctly emitted after checking if deposit was successful
29  +       emit GuardianAdded(msg.sender, token);
30
31          return address(tokenVault);
32      }
```

[L-8] Incorrect Name and Symbol when creating tokenVault for i_TokenTwo

Description: In `VaultGuardianBase::becomeTokenGuardian` when an user calls becomeTokenGuardian a new tokenVault is created for the token the guardian has chosen. There are two accepted tokens, i_tokenOne and i_tokenTwo, with different name and symbol. But the function assigns `TOKEN_ONE_VAULT_NAME` & `TOKEN_ONE_VAULT_SYMBOL` for both token vaults, which is misleading to off-chain indexers.

```
1  else if (address(token) == address(i_tokenTwo)) {
2              tokenVault = new VaultShares(
3                  IVaultShares.ConstructorData({
4                      asset: token,
5                      // Invalid NAME & SYMBOL assign for i_tokenTwo
6                      vaultName: TOKEN_ONE_VAULT_NAME,
7                      vaultSymbol: TOKEN_ONE_VAULT_SYMBOL,
8                      guardian: msg.sender,
9                      allocationData: allocationData,
10                     aavePool: i_aavePool,
11                     uniswapRouter: i_uniswapV2Router,
12                     guardianAndDaoCut: s_guardianAndDaoCut,
13                     vaultGuardians: address(this),
14                     weth: address(i_weth),
15                     usdc: address(i_tokenOne)
16                 })
17 )}
```

Impact: The core problem is that the VaultShares contract, which represents the user's staked position, will use this name and symbol when created.

1. User Confusion and Loss of Trust (Primary Impact)Misleading Information:

- A user who stakes LINK (represented by i_tokenTwo) will receive a receipt or see a token in their wallet named "USDC Vault Shares" (using TOKEN_ONE_VAULT_NAME and TOKEN_ONE_VAULT_SYMBOL).
- Trust and Error: This misalignment can cause panic, as the user might believe their funds were accidentally staked into the wrong asset vault, leading to support issues and a significant loss of trust in the protocol.

2. Broken Off-Chain Systems and Integration

- DEX/Aggregator Misidentification: If the protocol integrates with any DEX, portfolio tracker, or DeFi aggregator, those services rely on the token symbol and name to correctly identify the underlying asset. If the LINK vault shares are labeled as USDC vault shares, the tracking systems will display incorrect portfolio values and holdings.
- Indexing Issues: Off-chain services that index all deployed vaults will see two distinct vault addresses (one for USDC, one for LINK) but with identical metadata, making it impossible to programmatically distinguish between them without reading the vault's internal asset variable.

Recommended Mitigation: Use the correct name & symbol for i_tokenTwo.

```
else if (address(token) == address(i_tokenTwo)) {
            tokenVault = new VaultShares(
                IVaultShares.ConstructorData({
                    asset: token,
                    // Use the correct NAME & SYMBOL assign for
                       i_tokenTwo
+                   vaultName: TOKEN_TWO_VAULT_NAME,
+                   vaultSymbol: TOKEN_TWO_VAULT_SYMBOL,
                    guardian: msg.sender,
                    allocationData: allocationData,
                    aavePool: i_aavePool,
                    uniswapRouter: i_uniswapV2Router,
                    guardianAndDaoCut: s_guardianAndDaoCut,
                    vaultGuardians: address(this),
                    weth: address(i_weth),
                    usdc: address(i_tokenOne)
                })
)}
```

[L-9] Return value not assigned to named return variable in _aaveDivest

Description: In `AaveAdapter` `_aaveDivest` calls `withdraw` on the Aave pool to retrieve the invested amount. This withdraw call returns a uint256 value, the received amount. `_aaveDivest` has declared a return variable `amountOfAssetReturned`, but the Aave pool withdraw call return value it is never assigned to it. `_aaveDivest` will always return 0 in this case.

```
function _aaveDivest(
        IERC20 token,
        uint256 amount
    ) internal returns (uint256 amountOfAssetReturned) {
        i_aavePool.withdraw({
            asset: address(token),
            amount: amount,
            to: address(this)
        });
    }
```

Impact: The consequence of this is that the calling function receives a 0. Since the return value is not used by the caller, the integrity of the protocol's logic is not affected. It's purely a broken function implementation.

Recommended Mitigation: Assign the withdraw return value to the defined named return variable. Explicit return is optional due to the named return param, but it is good practice for clarity.

```
function _aaveDivest(
        IERC20 token,
        uint256 amount
    ) internal returns (uint256 amountOfAssetReturned) {
+       amountOfAssetReturned = i_aavePool.withdraw({
            asset: address(token),
            amount: amount,
            to: address(this)
        });
+       return amountOfAssetReturned;
    }
```

[L-10] Missing check for sufficient liquidity before calling removeLiquidity()

Description:

_uniswapDivest() calls i_uniswapRouter.removeLiquidity() using liquidityAmount without verifying that the vault holds at least that many LP tokens. In the current implementation, divestThenInvest() always passes the vault's full LP balance, preventing this from triggering. However, if _uniswapDivest is reused elsewhere or modified, it could cause unexpected reverts during divestment if the balance check is missing.

Impact:

Potential for failed divestment operations or locked liquidity in edge cases(rounding or future refactors).

Recommended Mitigation:

Add an explicit check in _uniswapDivest to ensure liquidityAmount <= i_uniswapLiquidityToken.balanceOf(address(this)

```
function _uniswapDivest(IERC20 token, uint256 liquidityAmount) internal
    returns (uint256 amountOfAssetReturned) {
        IERC20 counterPartyToken = token == i_weth ? i_tokenOne :
           i_weth;

        // Check LP balance first
+       uint256 lpBal = s_uniswapLP.balanceOf(address(this));
+       if (liquidityAmount == 0 || liquidityAmount > lpBal) {
+           revert("Invalid liquidity amount");
```

```
+       }

        (uint256 tokenAmount, uint256 counterPartyTokenAmount) =
            i_uniswapRouter.removeLiquidity({
            tokenA: address(token),
            tokenB: address(counterPartyToken),
            liquidity: liquidityAmount,
            amountAMin: 0,
            amountBMin: 0,
            to: address(this),
            deadline: block.timestamp
        });
        s_pathArray = [address(counterPartyToken), address(token)];
        uint256[] memory amounts = i_uniswapRouter.
            swapExactTokensForTokens({
            amountIn: counterPartyTokenAmount,
            amountOutMin: 0,
            path: s_pathArray,
            to: address(this),
            deadline: block.timestamp
        });
        emit UniswapDivested(tokenAmount, amounts[1]);
        amountOfAssetReturned = amounts[1];
    }
```

[L-11] Misleading/Incorrect event emision in updateGuardianAndDaoCut

Description: In `VaultGuardians::updateGuardianAndDaoCut` the event emitted doesn't correctly describe the function action. The function updates the Guardian and Dao fee, but the event is stating "UpdatedStakePrice".

```
/*
     * @notice Updates the percentage shares guardians & Daos get in
         new vaults
     * @param newCut the new cut
     * @dev this value will be divided by the number of shares whenever
         a user deposits into a vault
     * @dev historical vaults will not have their cuts updated, only
         future vaults
     */
    function updateGuardianAndDaoCut(uint256 newCut) external onlyOwner
        {
        s_guardianAndDaoCut = newCut;
        // Incorrect event title
        emit VaultGuardians__UpdatedStakePrice(s_guardianAndDaoCut,
            newCut);
    }
```

Impact: Causes misleading logs and confusion for off-chain monitoring.

Recommended Mitigation:

Consider using a correctly describing event title, like the already existing and unused event “Updated-Fee”.

```
function updateGuardianAndDaoCut(uint256 newCut) external onlyOwner {
        s_guardianAndDaoCut = newCut;
-       emit VaultGuardians__UpdatedStakePrice(s_guardianAndDaoCut, newCut);
+       event VaultGuardians__UpdatedFee(s_guardianAndDaoCut, newCut);
    }
```

[L-12] Ambiguous guardian and DAO cut calculation can lead to misleading configuration

Description:

In `VaultShares::deposit` the guardian and DAO fee is calculated using shares / i_guardianAndDaoCut, where i_guardianAndDaoCut is set to 1000. This treats the value as a divisor rather than a percentage or basis-point ratio. The naming (“cut”) suggests a proportional fee, which can confuse or lead to incorrect future updates. Additionally, the same fraction is minted twice — once for the guardian and once for the DAO — doubling the total effective fee.

```
_mint(i_guardian, shares / i_guardianAndDaoCut);
_mint(i_vaultGuardians, shares / i_guardianAndDaoCut);
```

Impact: The total protocol cut is ambiguous and may be misconfigured or misinterpreted, leading to unintended fee rates and inconsistent accounting across vaults.

Proof of Concept:

With i_guardianAndDaoCut = 1000 => each receives 0.1% of user shares, totaling 0.2%.

Recommended Mitigation:

Clarify the intended meaning of i_guardianAndDaoCut. If it is meant to represent a fee in basis points, use:

```
+   uint256 cutSharesTotal = (shares * i_guardianAndDaoCut) / 10_000;
+   uint256 cutShare = cutSharesTotal / 2;
    _mint(i_guardian, cutShares);
    _mint(i_vaultGuardians, cutShares);
```

Otherwise, rename it to guardianAndDaoDivisor and document that it is used as a divisor for clarity.

[L-13] Missing event emission on user deposit

Description:

The `deposit` function in `VaultShares` doesn't emit an event when the operation is completed, indicating that the deposit happened.. This function changes the state by minting vault shares to the user and shares cut to the DAO and guardian. Events should be emitted when state is changed.

While OpenZeppelin's ERC4626 emits a Deposit event for the main user deposit, the additional minting to the guardian and DAO happens silently.

Impact:

Off-chain systems and indexers cannot accurately track vault inflows or fee distributions, reducing transparency and potentially breaking accounting or analytics that rely on emitted events.

Recommended Mitigation:

Emit a dedicated event when shares are minted to the guardian and DAO, on deposit operations:

```
+   event GuardianAndDaoSharesMinted(address indexed guardian, address indexed dao, uint256 guardianShares, uint256 daoShares);

+   emit GuardianAndDaoSharesMinted(i_guardian, i_vaultGuardians, shares / i_guardianAndDaoCut, shares / i_guardianAndDaoCut);
```

This improves transparency and facilitates accurate monitoring of fee allocations.

Informational

[I-1] Unused interfaces `IInvestableUniverseAdapter` and `IVaultGuardians` with commented-out functions

Description:

1. The interface `IInvestableUniverseAdapter` defines commented-out function stubs and is not referenced anywhere in the codebase.
2. The interface `IVaultGuardians` has an empty body and is not referenced anywhere in the codebase.

Impact: No runtime impact, can increase audit complexity and create confusion

Recommended Mitigation: Consider removing unused files and data

File	% Lines	% Statements	% Branches	% Funcs
src/abstract/AStaticTokenData.sol	100.00% (4/4)	100.00% (2/2)	100.00% (0/0)	100.00% (2/2)
src/abstract/AStaticUSDCData.sol	100.00% (4/4)	100.00% (2/2)	100.00% (0/0)	100.00% (2/2)
src/abstract/AStaticWethData.sol	100.00% (4/4)	100.00% (2/2)	100.00% (0/0)	100.00% (2/2)
src/dao/VaultGuardianGovernor.sol	0.00% (0/6)	0.00% (0/4)	100.00% (0/0)	0.00% (0/3)
src/dao/VaultGuardianToken.sol	66.67% (4/6)	50.00% (2/4)	100.00% (0/0)	66.67% (2/3)
src/protocol/VaultGuardians.sol	100.00% (10/10)	100.00% (8/8)	100.00% (0/0)	100.00% (3/3)
src/protocol/VaultGuardiansBase.sol	95.65% (66/69)	93.55% (58/62)	81.82% (9/11)	100.00% (16/16)
src/protocol/VaultShares.sol	91.55% (65/71)	91.80% (56/61)	50.00% (4/8)	94.74% (18/19)
src/protocol/investableUniverseAdapters/AaveAdapter.sol	66.67% (6/9)	71.43% (5/7)	0.00% (0/1)	66.67% (2/3)
src/protocol/investableUniverseAdapters/UniswapAdapter.sol	61.54% (16/26)	61.29% (19/31)	0.00% (0/3)	66.67% (2/3)
Total	85.65% (179/209)	84.15% (154/183)	56.52% (13/23)	87.50% (49/56)

Figure 1: alt text

[I-2] Insufficient test coverage

Recommended Mitigation: Aim to get test coverage up to over 90% for all files and improve Branch testing.

[I-3] No check on return value in _aaveDivest

Description: In `AaveAdapter`, `_aaveDivest` defines a named return value which is never assigned(see L-9) and never checked

```
function _aaveDivest(
        IERC20 token,
        uint256 amount
    ) internal returns (uint256 amountOfAssetReturned) {
        i_aavePool.withdraw({
            asset: address(token),
            amount: amount,
            to: address(this)
        });

    }
```

Impact: There is minimal impact becuase the return value it is not used by other functions and Aave's `withdraw` is expected to revert on failure, mitigating the most severe risk.

Recommended Mitigation: Add a check on the return value of `_aaveDivest`

```
function _aaveDivest(
        IERC20 token,
        uint256 amount
    ) internal returns (uint256 amountOfAssetReturned) {
+       amountOfAssetReturned = i_aavePool.withdraw({
            asset: address(token),
            amount: amount,
```

```
            to: address(this)
        });
+       if (amountOfAssetReturned == 0) {
+           revert AaveAdapter__NothingToWithdraw();
+       }
+       return amountOfAssetReturned;
    }
```

Alternatively, remove the unused return value from the function signature to avoid confusion.

[I-4] Typo in function title in getUniswapLiquidtyToken

Description:

In `VaultShares` the function title contains a misspelled word(missing an "i"): Liquidty, instead of Liquidity

Impact: No runtime impact, can create confusion

Recommended Mitigation: Consider correcting the function title to `getUniswapLiquidityToken`

```
+   function getUniswapLiquidityToken() external view returns (address)
    {
        return address(i_uniswapLiquidityToken);
    }
```

[I-5] Missing mechanism for adding new investment platforms

Description:

The README file states that the project can add new investment strategies, but there is no function or mechanism to allow this.

"The protocol is upgradeable so that if any of the platforms in the investable universe change, or we want to add more, we can do so."

In the current version, there is only the posibility to update the investment strategy/allocation.

Impact: Misleading description of the protocol abilities.

Recommended Mitigation: Consider removing the part that states that the protocol can add new investment platforms or implement a mechanism that allows new investment platforms to be safely added.

Gas

[G-1] Functions marked `public` are not used internally

Description: A function is declared as **`public`** when it can be declared as external. The **`public`** visibility forces the compiler to generate code that copies all function arguments from calldata to memory (specifically, into the free memory pointer region) when the function is called.

2 Found Instances

- Found in src/protocol/VaultShares.sol Line: 115

```
1       function setNotActive() public onlyVaultGuardians isActive {
```

- Found in src/protocol/VaultShares.sol Line: 181

```
1       function rebalanceFunds() public isActive divestThenInvest
           nonReentrant {}
```

This copying operation generates a significant gas consumption for every function call. When the function is only intended to be called by external accounts (EOAs) or other contracts, and not internally by other functions within the same contract.

Impact: Using **`public`** leads to waste of gas, directly increasing the transaction cost for the user.

Recommended Mitigation: Change visibilty from **`public`** to `external`, if not used internally.